

# An Introduction to Creating Multi-Sheet Microsoft Excel Workbooks the Easy Way with SAS®

Vince DelGobbo, SAS Institute Inc, Cary, NC

## REQUIREMENTS

(1) Base SAS Software, 9.1.3 or later, *any operating system*, (2) **recent version** of the ExcelXP ODS tagset (see the section Additional Information below), (3) Microsoft Excel 2002 (aka Excel XP) or later.

## GENERAL STEPS

(1) Run SAS code to create output, (2) open output with Excel, (3) make changes to the SAS code to add the features you want.

## BASIC CODE

```
ods tagsets.ExcelXP file='MyWorkbook.xml' style=sansPrinter options(name='value');  
  * Your SAS procedure code here;  
ods tagsets.ExcelXP close;
```

## ODS TAGSET OPTIONS - SPECIFIED ON THE ODS STATEMENT

SHEET\_NAME: Default value: 'None'; Values: 'None', string.

EMBEDDED\_FOOTNOTES, EMBEDDED\_TITLES: Default value: 'No'; Values: 'Yes', 'No'.

PRINT\_HEADER, PRINT\_FOOTER: Default value: "; Values: string.

Example: &C&A&RPage &P of &N

Left-justify Text	&L	Center text	&C
Right-justify text	&R	Insert sheet name	&A
Insert page number	&P	Insert number of pages	&N
Insert date printed	&D	Insert time printed	&T
Insert file name	&F	Insert file path	&Z

AUTOFILTER: Default value: 'None'; Values: 'None', 'All', range.

## SAS SYSTEM OPTIONS - SPECIFIED ON THE SAS "OPTIONS" STATEMENT

papersize: Default value: 'Letter'; Values: string. See below for commonly-used values.

Option Value	Paper Size	Option Value	Paper Size
Letter	8 1/2" x 11"	Legal	8 1/2" x 14"
Tabloid	11" x 17"	Ledger	17" x 11"
A4	210mm x 297mm	A3	297mm x 420mm

topmargin, bottommargin, leftmargin, rightmargin: Values: string.

Example: options topmargin=1 in bottommargin=1 in leftmargin=0.5 in rightmargin=0.5 in;

## ODS STYLE OVERRIDES

(1) Find an ODS style you like, (2) make a copy, (3) make modifications to the copy, (4) use the new style, and specify style overrides.

Only the PRINT, REPORT and TABULATE procedures support this technique.

# An Introduction to Creating Multi-Sheet Microsoft Excel Workbooks the Easy Way with SAS®

Vince DelGobbo, SAS Institute Inc, Cary, NC

## ODS XLSANSPRINTER STYLE CODE FRAGMENT

```
style data_male from data / background=#99CCFF;

style data_female from data / background=#FF99CC;

style data_male_d1 from data_male / tagattr='format:#.0';

style data_female_d1 from data_female / tagattr='format:#.0';
```

## FINAL SAS CODE

```
ods tagsets.ExcelXP path='some-directory' file='MyWorkbook.xml' style=XLSansPrinter;

title 'The CLASS Dataset';
footnote '(From the SASHELP library)';

ods tagsets.ExcelXP options(embedded_titles='yes' embedded_footnotes='yes'
                             print_header='&C&A&RPage &P of &N'
                             print_footer='&RPrinted &D at &T'
                             autofilter='2');

ods tagsets.ExcelXP options(sheet_name='Male Students');

proc print data=sashelp.class noobs;
  where (sex eq 'M');
  var name age / style(Column)=data_male;
  var height weight / style(Column)=data_male_d1;
run; quit;

ods tagsets.ExcelXP options(sheet_name='Female Students');

proc print data=sashelp.class noobs;
  where (sex eq 'F');
  var name age / style(Column)=data_female;
  var height weight / style(Column)=data_female_d1;
run; quit;

ods tagsets.ExcelXP close;
```

## ADDITIONAL INFORMATION

A recent version of the ExcelXP ODS tagset is **required**, and can be downloaded from:  
<http://support.sas.com/rnd/base/ods/odsmarkup/excltags.tpl>

Refer to page 4 of this paper for important information on how to install and use the tagset:  
"More Tips and Tricks for Creating Multi-Sheet Microsoft Excel Workbooks the Easy Way with SAS"  
[Available at: http://support.sas.com/rnd/papers/index.html#excel2009](http://support.sas.com/rnd/papers/index.html#excel2009)

"Reporting Procedure Styles Tip Sheet" contains information about using ODS style overrides.  
[Available at: http://support.sas.com/rnd/base/ods/scratch/reporting-styles-tips.pdf](http://support.sas.com/rnd/base/ods/scratch/reporting-styles-tips.pdf)

If your registered in-house or local SAS users group would like to request a presentation as your annual SAS presentation (as a seminar, talk, or workshop) at an upcoming meeting, please submit an online User Group Request Form ([support.sas.com/usergroups/namerica/lug-form.html](http://support.sas.com/usergroups/namerica/lug-form.html)) at least eight weeks in advance.